Quiz 1 *Name:* _____

9/24/12

Examine the code below and tell us the result that will be printed by each of the four `println()`s found in `main()`. For full credit, outline your reasoning for each answer.

```
1  class Test {
2      def f(x:Long):Long = x < 2 ? 1+x : f(x-1)*f(x-2);
3
4      def g(m:Long):Long {
5          var result:Long=2L, a:Long=1L;
6          for (y in 2L..m) {
7              val b=result;
8              result *=a;
9              a=b;
10         }
11         return result;
12     }
13
14     def p(m:Long):Long {
15         var result:Long=2L, a:Long=1L;
16         finish for (y in 2L..m) async {
17                 val b=result;
18                 atomic {
19                     result *=a;
20                     a=b;
21                 }
22             }
23         return result;
24     }
25
26     def a(m:Long):Long {
27         val result=new Cell[Long](2L), a=new Cell[Long](1L);
28         val P=Place.MAX_PLACES;
29         finish for (y in 2L..m) at (Place((y%P) as Int)) async {
30                 val b=result();
31                 atomic {
32                     result() *=a();
33                     a()=b;
34                 }
35             }
36         return result();
37     }
38
39     public static def main(args:Rail[String]) {
40         val x = new Test();
41         val v= args.size>0? Long.parseLong(args(0)):5L;
42         val ans=x.f(v);
43         Console.OUT.println(ans);          // Line 0
44         Console.OUT.println(ans==x.g(v));  // Line 1
45         Console.OUT.println(ans==x.p(v));  // Line 2
46         Console.OUT.println(ans==x.a(v));  // Line 3
47     }
48 }
```

Line 0 = 32
$f(0) = 1$
$f(1) = 2$
$f(2) = 2 \cdot 1 = 2$
$f(3) = 2 \cdot 2 = 4$
$f(4) = 2 \cdot 4 = 8$
$f(5) = 8 \cdot 4 = 32$

Line 1 = true
This iterative serial code performs the same computation as the recursive serial code in `f()`.

Line 2 = ?
This parallel implementation of the iterative algorithm will compute the same as the serial when the loop bodies are executed atomically. However, this is not guaranteed as only part of the body is actually atomic. Therefore, this code is nondeterministic, with some interleavings yielding 32 and others not.

Line 3 = false
This multiplace implementation *will not* return the correct result. To begin with, the result from Place 0 is returned – results from other places are never shared back. Moreover, the values will be incorrect in many places as the code assumes global references to `a` and `result` which are not `GlobalRefs`. Finally the atomicity issue that was present in the previous part remains in this one.